
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test cases 8.6.6.3 which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2013-03_D13k49’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution logs are provided as evidence.

4) Corrections required for test case 8.6.6.3
Introduction

This section describes the changes required to make test 8.6.6.3 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2013-03_D13wk49 release.

4.1 Change 1
	Function name
	f_TC_8_6_6_3_EUTRA

	Reason for change
	1. In steps 15-20 after the T3 cell configuration applies, UE is expected to perform reselection between cells 2 and 12 which have different PLMNs and are on different frequencies. Since the PLMN of cell2 is HPLMN on the default USIM and PLMN of the cell12 does not belong to EHLPMN, according to 23.122, cl. 4.4.3.1.1, UE is not obliged to perform PLMN search and may not discover cell12 even if it is stronger than current serving serving cell as long as the serving cell is suitable.

For this reason we propose to add PLMN of the cell12 into list of EPLMNs during the attach procedure in the test preamble.

Similarly, the PLMN of cell2 (same as cell1 PLMN) needs to be added in the list EPLMNs via the update procedure on cell12 during the steps 15-20, so the UE will perform reselection after step 22 into cell 1. 

Prose CR has been raised for this change.

2. According to 36.521-3, cl. 8.6.6.3.3.1 the c3 system information combination should be used in all the cells. C3 combination initialises SIB5 with the information required for inter-frequency transitions. However cell12 doesn’t belong to the list of cells with SIB5 according to 36.508, cl. 6.3.1.2. For this reason we propose to use cell 3 which belongs to that list and is on different frequency than cell 1 and 2. At the beginning of the test the PLMN of the cell 3 will be changed to different from PLMN of cells 1 and 2 so it will be in accordance with test purpose.

Prose CR has been raised for this change.

3. Due to change 1, the preamble needs to use function which has EPLMN list as an input parameter. As a result also a definition for T0 attenuation levels needs to be added in order to turn on the serving cell on which the preamble with EPLMN list is going to be performed.
4. In step 5, the HO should be from cell1 to cell2 and not to cell1 as per Table: 8.6.6.3.3.3-6.
5. In order to achieve HO failure after the step 5 the expiry of T304 has to happen prior configuration of cell1 to normal PRACH procedure otherwise UE will be able to re-establish prior the T304 which will not classify for HO failure but for other failure.
6. In step 9 the reestablishment needs to use rel9 template instead of the rel10 so it can accommodate situation where UE has no higher release IE to include. 
7. In step 10 the reconfiguration is incorrectly sent on cell1 instead of cell2.

8. Prior the steps 23-24 the security and bearers have to be restarted on cell1 in order to handle new connection to cell properly after unsuccessful HO earlier in the test.

9. The last function prior the postamble which is checking that UE is in Idle state in cell1 releases the RRC connection as a result. For this reason the subsequent postamble needs to be called from Idle and from Connected state.

	Summary of change
	1. . Content of the attach accept in test preamble has been changed to include PLMN of cell12 in the list of EPLMNs and content of TAU accept to include PLMN of cell1 in the list of EPLMNs 

2. Cell 12 has been replaced with cell 3 and PLMN of the cell3 has been changed to the one the cell12 would have had.
3.Preamble function has been replaced by function which lets to add list of EPLMNs and power level for serving cell has been set to T0 prior the preamble.

4. In step 5 the cellId has been corrected.
5. Configuring cell1 to norma PRACH operation mode has been moved after the T304 expiry
6. In step 9 the reestablishment rel9 template has been used instead of the rel10 one.
 7. In step 10 the cellId has been corrected.
8. Security and bearers have been restarted on cell1 prior new connection in steps 23-24.

9. The postamble is called from E1_Idle state instead of E2_Connected.

	TTCN module
	RRC_LoggedHOFailure.ttcn


Before change

      function f_TC_8_6_6_3_EUTRA() runs on EUTRA_PTC

  { //Handover Failure logging / Reporting of HOF report availability / PLMN change

    var EUTRA_CellInfo_Type v_CellInfo_Cell1,v_CellInfo_Cell12;
    var NAS_KsiValue v_KsiValue;

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(NORMAL);

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType();

    var NAS_MSG_Indication_Type v_NasInd ;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1,v_CellPowerList_AtT2,v_CellPowerList_AtT3,v_CellPowerList_AtT4;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var Frequency_IE_Type v_Frequency_IE_f1;

    var PhysCellId v_PhysCellId_Cell1,v_PhysCellId_Cell2;

    var float v_T304;

    var ShortMAC_I v_ShortMAC_I;

    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (omit) MobilityControlInfo v_MobilityControlInfo;

    var EUTRA_SecurityParams_Type v_AuthParams;

    timer t_T304Max;

    v_CellPowerList_AtT1 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell2, -79),

      cs_CellPower(eutra_Cell12, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, -79),

      cs_CellPower(eutra_Cell12, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_AtT3 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell12, -73)

    };

    v_CellPowerList_AtT4 := {

      cs_CellPower(eutra_Cell1, -73),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell12, tsc_ServingCellRS_EPRE)

    };

    // Init variables

    f_EUTRA_Init(c3);

    f_EUTRA_CellInfo_SetPreambleTransMax(eutra_Cell1, n50);

    v_T304 := f_EUTRA_SetTimerToleranceMax(eutra_Cell2, rrcTimer, 1.0);

    v_CellInfo_Cell12 := f_EUTRA_CellInfo_Get(eutra_Cell12);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);

    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth);

    //Set EPRE for RF Initial

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell1, -73);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell12, -73);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell12);

    // Preamble

    f_EUTRA_Preamble(eutra_Cell1, STATE2_IDLEUPDATE);
    f_EUTRA_RbEst_Def(eutra_Cell1);

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to setup intra-frequency measurement on Cell 1.

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_1Entry_IntraLTE(v_RRC_TI,

                                                                                                    v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                    v_ChBandDependency_Cell2.AllowedMeasBandwidth));

    //@siclog "Step 3" siclog@

    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog@

    //The UE transmits a MeasurementReport message on Cell 1.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,

                                    cr_MeasurementReport_Eutra_1Entry_both(1, v_PhysCellId_Cell2)));

    //Configure Cell2 to not send response on RACH preamble reception on Cell 2

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell2,

                                   omit,

                                   cs_RachProcedureConfig_NoResponse(f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell2)));

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including mobilityControlInfo on Cell 1.

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO(eutra_Cell1,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  cs_RACH_ConfigDedicated(63));

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1, cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI, omit, v_MobilityControlInfo,

                                                                        f_Generate_cs_508_RadioResourceConfigDedicated_HO (eutra_Cell2, n4),

                                                                        cs_508SecurityConfigHO_IntraLTE)));

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //EXCEPTION: In parallel to the events described in step 6 the steps specified in Table 8.6.4.4.3.2-3 should take place.

    //EXCEPTION: The steps 1 and 2 below are repeated for the duration of T304.

    //1. The UE attempts to perform the intra frequency handover using MAC Random Access Preamble on Cell 2.

    //2. The SS does not respond.

    t_T304Max.start(v_T304);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE in Step 7

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell2 );

    //Configure Cell1 to normal PRACH operation mode

    f_EUTRA_SS_ConfigRachProcedure( eutra_Cell2,

                                    tsc_C_RNTI_Def,

                                    cs_RachProcedureConfig_Def(omit,

                                                               f_EUTRA_CellInfo_GetRAR_TA( eutra_Cell2 ),

                                                               f_EUTRA_CellInfo_GetDL_ChBandwidth ( eutra_Cell2)));

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1 parameter according to the row "T2" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);

    t_T304Max.timeout;
    //@siclog "Step 7" siclog@

    //The UE transmits an RRCConnectionReestablishmentRequest message on Cell 2.

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell2,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysCellId_Cell1,

                                                                               cr_ReestablishmentCause_HandoverFailure,

                                                                               v_ShortMAC_I)));

    v_AuthParams := f_EUTRA_Security_Get ( );

    v_AuthParams := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell2, v_AuthParams, 0 );

    f_EUTRA_Security_Set ( v_AuthParams );

    //@siclog "Step 8" siclog@

    //The SS transmits an RRCConnectionReestablishment message on Cell 2.

    f_EUTRA_RRC_RRCConnectionReestablishment_Def(eutra_Cell2);

    //@siclog "Step 9" siclog@

    //The UE transmits an RRCConnectionReestablishmentComplete message with handover failure information on Cell 2.

    SRB.receive(car_SRB1_RrcPdu_IND (eutra_Cell2,

                                     cr_RRCConnectionReestablishmentComplete_r10(tsc_RRC_TI_Def, -, true_)));

    //@siclog "Step 10" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 2

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1, cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI, omit, omit,

                                                                        f_Generate_cs_508_RadioResourceConfigDedicated_HO (eutra_Cell2, n4),

                                                                        omit)));

    //@siclog "Step 11" siclog@

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 2.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionRelease message on Cell 2.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell2);

    //@siclog "Step 13" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 14" siclog@

    //The SS changes Cell 2 and Cell 12 parameters according to the row "T3" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 15-16" siclog@

    //Steps 1 to 2 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 12.

    //The UE transmits an RRCConnectionRequest message on the cell specified in the test case.

    //SS transmit an RRCConnectionSetup message.

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell12);

    //send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_Cell12);

    //@siclog "Step 17" siclog@

    //Check: Does the UE transmit an RRCConnectionSetupComplete message without handover failure information on Cell 12?

    //This message includes a TRACKING AREA UPDATE REQUEST message.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell12,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.6.3 Step 17");

    //@siclog "Step 18-20" siclog@

    //Steps 4 to 6 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 12.

    //Note: The UE performs a TAU procedure and the RRC connection is released.

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(eutra_Cell12, v_CellInfo_Cell12.NAS_Parameters.Guti_Parameters, v_EpsUpdate_TypeValue, v_EPS_ContextStatus, v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, NORMAL);

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell12);
    //@siclog "Step 21" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 22" siclog@

    //The SS changes Cell 1, Cell 2 and Cell 12 parameters according to the row "T4" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT4);

    //@siclog "Step 23-24" siclog@

    //Steps 1 to 2 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 1.

    //SS transmit an RRCConnectionSetup message.

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

    //send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell1);

    //@siclog "Step 25" siclog@

    //Check: Does the UE transmit an RRCConnectionSetupComplete message with handover failure information on Cell 1?

    //This message includes a TRACKING AREA UPDATE REQUEST message.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?, -, -, native, true_, omit, omit),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.6.3 Step 25");

    //@siclog "Step 26-28" siclog@

    //Steps 4 to 6 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 1.

    //Note: The UE performs a TAU procedure and the RRC connection is released.

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(eutra_Cell1, v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters, v_EpsUpdate_TypeValue, v_EPS_ContextStatus, v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, NORMAL);

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    //@siclog "Step 29" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 30" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.2 indicate that the UE is in E-UTRA RRC_IDLE state on Cell 1?

    f_EUTRA_508Check_IdleState (eutra_Cell1);

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

 } // end of f_TC_8_6_6_3_EUTRA
After change

        function f_TC_8_6_6_3_EUTRA() runs on EUTRA_PTC

  { //Handover Failure logging / Reporting of HOF report availability / PLMN change

    var EUTRA_CellInfo_Type v_CellInfo_Cell1,v_CellInfo_Cell3;
    var NAS_KsiValue v_KsiValue;

    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(NORMAL);

    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType();

    var NAS_MSG_Indication_Type v_NasInd ;

    var template (value) CellPowerList_Type v_CellPowerList_AtT0,v_CellPowerList_AtT1,v_CellPowerList_AtT2,v_CellPowerList_AtT3,v_CellPowerList_AtT4;

    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus;

    var Frequency_IE_Type v_Frequency_IE_f1;

    var PhysCellId v_PhysCellId_Cell1,v_PhysCellId_Cell2;

    var float v_T304;

    var ShortMAC_I v_ShortMAC_I;

    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell2;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (omit) MobilityControlInfo v_MobilityControlInfo;

    var EUTRA_SecurityParams_Type v_AuthParams;

    timer t_T304Max;

    v_CellPowerList_AtT0 := {

        cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE)

    };
    v_CellPowerList_AtT1 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell2, -79),

      cs_CellPower(eutra_Cell3, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, -79),

      cs_CellPower(eutra_Cell3, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_AtT3 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE),

      cs_CellPower(eutra_Cell3, -73)

    };

    v_CellPowerList_AtT4 := {

      cs_CellPower(eutra_Cell1, -73),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell3, tsc_ServingCellRS_EPRE)

    };

    // Init variables

    f_EUTRA_Init(c3);

    f_EUTRA_CellInfo_SetPLMN_1Entry ( eutra_Cell3, cs_HPLMN_002_11);
    f_EUTRA_CellInfo_SetPreambleTransMax(eutra_Cell1, n50);

    v_T304 := f_EUTRA_SetTimerToleranceMax(eutra_Cell2, rrcTimer, 1.0);

    v_CellInfo_Cell3 := f_EUTRA_CellInfo_Get(eutra_Cell3);

    v_CellInfo_Cell1 := f_EUTRA_CellInfo_Get(eutra_Cell1);

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell2);

    v_PhysCellId_Cell1 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1);

    v_PhysCellId_Cell2 := f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell2);

    v_ChBandDependency_Cell2 := f_EUTRA_BandDependentParam(v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth);

    //Set EPRE for RF Initial

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell1, -73);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell2, -79);

    f_EUTRA_CellInfo_InitMaxReferencePower(eutra_Cell3, -73);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    f_EUTRA_CellConfig_Def(eutra_Cell3);

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT0);
    // Preamble

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);

    f_EUTRA_IdleUpdated_WithEplmn (eutra_Cell1,

                                   PREAMBLE,

                                   cs_PLMN_List_1PLMN (f_Asn2Nas_PlmnId (f_EUTRA_CellInfo_GetGutiPLMN (eutra_Cell3))));
    f_EUTRA_RbEst_Def(eutra_Cell1);

    v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 1-2" siclog@

    //The SS transmits an RRCConnectionReconfiguration message to setup intra-frequency measurement on Cell 1.

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 1.

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_RRCConnectionReconfiguration_1Entry_IntraLTE(v_RRC_TI,

                                                                                                    v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq,

                                                                                                    v_ChBandDependency_Cell2.AllowedMeasBandwidth));

    //@siclog "Step 3" siclog@

    //The SS changes Cell 1 and Cell 2 parameters according to the row "T1" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 4" siclog@

    //The UE transmits a MeasurementReport message on Cell 1.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1,

                                    cr_MeasurementReport_Eutra_1Entry_both(1, v_PhysCellId_Cell2)));

    //Configure Cell2 to not send response on RACH preamble reception on Cell 2

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell2,

                                   omit,

                                   cs_RachProcedureConfig_NoResponse(f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell2)));

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including mobilityControlInfo on Cell 1.

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO(eutra_Cell2,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  -,

                                                                  cs_RACH_ConfigDedicated(63));

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1, cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI, omit, v_MobilityControlInfo,

                                                                        f_Generate_cs_508_RadioResourceConfigDedicated_HO (eutra_Cell2, n4),

                                                                        cs_508SecurityConfigHO_IntraLTE)));

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_PUCCH_Synch_None));

    //EXCEPTION: In parallel to the events described in step 6 the steps specified in Table 8.6.4.4.3.2-3 should take place.

    //EXCEPTION: The steps 1 and 2 below are repeated for the duration of T304.

    //1. The UE attempts to perform the intra frequency handover using MAC Random Access Preamble on Cell 2.

    //2. The SS does not respond.

    t_T304Max.start(v_T304);

    //Calculate ShortMAC-I to be received in RRCConnectionReestablishment Request from UE in Step 7

    v_ShortMAC_I := f_Calculate_ShortMAC ( eutra_Cell1, eutra_Cell2 );

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1 parameter according to the row "T2" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT2);
    t_T304Max.timeout;

    //Configure Cell1 to normal PRACH operation mode

    f_EUTRA_SS_ConfigRachProcedure( eutra_Cell2,

                                    tsc_C_RNTI_Def,

                                    cs_RachProcedureConfig_Def(omit,

                                                               f_EUTRA_CellInfo_GetRAR_TA( eutra_Cell2 ),

                                                               f_EUTRA_CellInfo_GetDL_ChBandwidth ( eutra_Cell2)));

    //@siclog "Step 7" siclog@

    //The UE transmits an RRCConnectionReestablishmentRequest message on Cell 2.

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_Cell2,

                                    cr_508_RRCConnectionReestablishmentRequest(tsc_C_RNTI_Def,

                                                                               v_PhysCellId_Cell1,

                                                                               cr_ReestablishmentCause_HandoverFailure,

                                                                               v_ShortMAC_I)));

    v_AuthParams := f_EUTRA_Security_Get ( );

    v_AuthParams := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( eutra_Cell2, v_AuthParams, 0 );

    f_EUTRA_Security_Set ( v_AuthParams );

    //@siclog "Step 8" siclog@

    //The SS transmits an RRCConnectionReestablishment message on Cell 2.

    f_EUTRA_RRC_RRCConnectionReestablishment_Def(eutra_Cell2);

    //@siclog "Step 9" siclog@

    //The UE transmits an RRCConnectionReestablishmentComplete message with handover failure information on Cell 2.

    SRB.receive(car_SRB1_RrcPdu_IND (eutra_Cell2,

                                     cr_RRCConnectionReestablishmentComplete_RLF_r9(tsc_RRC_TI_Def)));

    //@siclog "Step 10" siclog@

    //The SS transmits an RRCConnectionReconfiguration message on Cell 2

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell2, cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI, omit, omit,

                                                                        f_Generate_cs_508_RadioResourceConfigDedicated_HO (eutra_Cell2, n4),

                                                                        omit)));

    //@siclog "Step 11" siclog@

    //The UE transmits an RRCConnectionReconfigurationComplete message on Cell 2.

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionRelease message on Cell 2.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell2);

    //@siclog "Step 13" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 14" siclog@

    //The SS changes Cell 2 and Cell 12 parameters according to the row "T3" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 15-16" siclog@

    //Steps 1 to 2 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 12.

    //The UE transmits an RRCConnectionRequest message on the cell specified in the test case.

    //SS transmit an RRCConnectionSetup message.

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell3);

    //send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_Cell3);

    //@siclog "Step 17" siclog@

    //Check: Does the UE transmit an RRCConnectionSetupComplete message without handover failure information on Cell 12?

    //This message includes a TRACKING AREA UPDATE REQUEST message.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell3,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.6.3 Step 17");

    //@siclog "Step 18-20" siclog@

    //Steps 4 to 6 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 12.

    //Note: The UE performs a TAU procedure and the RRC connection is released.

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(eutra_Cell3, v_CellInfo_Cell3.NAS_Parameters.Guti_Parameters, v_EpsUpdate_TypeValue, v_EPS_ContextStatus, v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, NORMAL, cs_PLMN_List_1PLMN(f_Asn2Nas_PlmnId(f_EUTRA_CellInfo_GetGutiPLMN(eutra_Cell2))));
    f_EUTRA_RRC_ConnectionRelease(eutra_Cell3);

    //@siclog "Step 21" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 22" siclog@

    //The SS changes Cell 1, Cell 2 and Cell 12 parameters according to the row "T4" in Table 8.6.6.3.3.2-1.

    f_EUTRA_SetCellPowerList(v_CellPowerList_AtT4);

    f_EUTRA_SS_RRC_ReleaseSecurity(eutra_Cell1, cs_TimingInfo_Now);  

    f_EUTRA_SS_SRBs_DRBs_Release(eutra_Cell1, cs_TimingInfo_Now);

    f_EUTRA_SS_SRBs_DRBs_Config(eutra_Cell1, cs_TimingInfo_Now);
    //@siclog "Step 23-24" siclog@

    //Steps 1 to 2 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 1.

    //SS transmit an RRCConnectionSetup message.

    // receive RRC Connection Request on SRB0 (CCCH):

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

    //send RRC Connection Setup on SRB0 (CCCH) and switch on UL grant assignments:

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell1);

    //@siclog "Step 25" siclog@

    //Check: Does the UE transmit an RRCConnectionSetupComplete message with handover failure information on Cell 1?

    //This message includes a TRACKING AREA UPDATE REQUEST message.

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,

                                                       cr_RRCConnectionSetupComplete_r10 (tsc_RRC_TI_Def, ?, -, -, native, true_, omit, omit),

                                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                         cr_508_TAU_Request(v_EpsUpdate_TypeValue,

                                                                                            v_KsiValue,

                                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                                            v_AdditionalUpdateType_Expected)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.6.3 Step 25");

    //@siclog "Step 26-28" siclog@

    //Steps 4 to 6 of the generic test procedure in TS 36.508 [18] subclause 6.4.2.7 are performed on Cell 1.

    //Note: The UE performs a TAU procedure and the RRC connection is released.

    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    f_EUTRA_TrackingAreaUpdate_Accept_Complete(eutra_Cell1, v_CellInfo_Cell1.NAS_Parameters.Guti_Parameters, v_EpsUpdate_TypeValue, v_EPS_ContextStatus, v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, NORMAL);

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    //@siclog "Step 29" siclog@

    //Wait for 5 s for the UE to enter E-UTRA RRC_IDLE state.

    f_Delay(5.0);

    //@siclog "Step 30" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 subclause 6.4.2.2 indicate that the UE is in E-UTRA RRC_IDLE state on Cell 1?

    f_EUTRA_508Check_IdleState (eutra_Cell1);

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);

 } // end of f_TC_8_6_6_3_EUTRA
5) Execution Log Files

Qualcomm UE
The Qualcomm MDM9625 UE passed this test case on the Anritsu ME7832L LTE system with NAS and AS Snow 3G integrity and ciphering algorithm in FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

6 References

	[1]
	R5s131038: This archive comprises text format execution log file.

